New Stable Periodic Solutions to the $n$-Body Problem

Robert J. Vanderbei

2014 Feb 17

Program in Applied and Computational Mathematics
Princeton University

http://www.princeton.edu/~rvdb
Figure Eight 3, 4, 5 Bodies
Exotic But Unstable
New Ones
minimize $f(x)$
subject to $b \leq h(x) \leq b + r,$
      $l \leq x \leq u$

- **Linear Programming (LP):** $f$ and $h$ are linear.
- **Convex Optimization:** $f$ is convex, each $h_i$ is concave, and $r = \infty$.
- **Nonlinear Optimization:** $f$ and each $h_i$ is assumed to be twice differentiable

- Generally, we seek a *local solution* in the vicinity of a given starting point.
- If problem is convex (which includes LP), any local solution is automatically a *global solution*.

Least Action Principle

Given: $n$ bodies.

Let:

- $m_j$ denote the mass and
- $z_j(t)$ denote the position in $\mathbb{R}^2 = \mathbb{C}$ of body $j$ at time $t$.

Action Functional:

$$A = \int_0^{2\pi} \left( \sum_j \frac{1}{2} m_j \| \dot{z}_j \|^2 + \sum_{j,k:k<j} \frac{G m_j m_k}{\| z_j - z_k \|} \right) dt.$$
Equation of Motion

First Variation:

\[
\delta A = \int_0^{2\pi} \sum_\alpha \left( \sum_j m_j \dot{z}_j^\alpha \delta \dot{z}_j^\alpha - \sum_{j,k:k<j} G m_j m_k \frac{(z_j^\alpha - z_k^\alpha)(\delta z_j^\alpha - \delta z_k^\alpha)}{||z_j - z_k||^3} \right) dt
\]

\[
= - \int_0^{2\pi} \sum_j \sum_\alpha \left( m_j \ddot{z}_j^\alpha + \sum_{k:k\neq j} G m_j m_k \frac{z_j^\alpha - z_k^\alpha}{||z_j - z_k||^3} \right) \delta z_j^\alpha dt
\]

Setting first variation to zero, we get:

\[
m_j \ddot{z}_j^\alpha = - \sum_{k:k\neq j} G m_j m_k \frac{z_j^\alpha - z_k^\alpha}{||z_j - z_k||^3}, \quad j = 1, 2, \ldots, n, \quad \alpha = 1, 2
\]

Note: If \( m_j = 0 \) for some \( j \), then the first order optimality condition reduces to \( 0 = 0 \), which is not the equation of motion for a massless body.
Consider two bodies of equal mass $m$ on a circular orbit of radius $r$:

$$z(t) = re^{i\omega t \pm \pi/2}.$$  

Differentiating twice, we get

$$\ddot{z}(t) = -r\omega^2 e^{i\omega t \pm \pi/2}.$$  

From Newton’s laws of motion, we get

$$-r\omega^2 e^{i\omega t \pm \pi/2} = \ddot{z}(t) = -\frac{Gm}{(2r)^2} e^{i\omega t \pm \pi/2}.$$  

Solving for $r$, we get

$$r = \left(\frac{Gm}{4\omega^2}\right)^{1/3}.$$
Double/Double Style Initialization

Start with a pair of bodies each of mass $2m$ in a circular orbit:

$$z(t) = \pm r_0 e^{i\omega_0 t}$$

where

$$r_0 = \left( \frac{2Gm}{4\omega_0^2} \right)^{1/3}.$$

Then assume that each mass is actually a 2-body system in a circular orbit with radius $r_1$ with given initial phase shifts:

$$z(t) = \pm r_0 e^{i\omega_0 t} \pm r_1 e^{i\omega_1 t + i\phi_\pm}.$$

Here,

$$r_1 = \left( \frac{Gm}{4\omega_1^2} \right)^{1/3}. $$
Hierarchies of 2-Body Systems

Double/Double

- Sun
- Sun 2
- Earth
- Moon

Mini/Month/Year

- Sun
- Earth
- Moon
- Minimoon
param N := 4; # number of masses
param M := 40000; # number of terms in numerical approx to integral
param G := 1;
param period := 2*pi/omega0; # temporal length of the orbit segment
param dt := period / M;

var x {j in 0..N-1, i in 0..M-1} >= -5, <= 5;
var y {j in 0..N-1, i in 0..M-1} >= -5, <= 5;

var xdot {j in 0..N-1, i in 0..M-1} = (x[j,(i+1) mod M]-x[j,i])/dt;
var ydot {j in 0..N-1, i in 0..M-1} = (y[j,(i+1) mod M]-y[j,i])/dt;

var K {i in 0..M-1}
    = 0.5 * sum {j in 0..N-1} (xdot[j,i]^2 + ydot[j,i]^2);

var P {i in 0..M-1}
    = - sum {j in 0..N-1, k in 0..N-1: k<j}
      1/sqrt((x[j,i]-x[k,i])^2 + (y[j,i]-y[k,i])^2);

minimize Action: sum {i in 0..M-1} (K[i] - P[i])*dt;
A Double/Double Initialization

\[
\begin{align*}
\text{param } \omega_0 & := 1./2; \\
\text{param } \omega_1 & := 1./1; \\
\text{param } \phi_p & := -\pi/2; \\
\text{param } \phi_m & := \pi/2; \\
\text{param } r_0 & := (G/(2*\omega_0^2))^{1./3.}; \\
\text{param } r_1 & := (G/(4*\omega_1^2))^{1./3.}; \\
\text{param } t \{i \in 0..M-1\} & := i*\text{dt};
\end{align*}
\]

\[
\begin{align*}
\text{let } \{k \in 0..M-1\} x[0,k] & := r_0\cos(\omega_0*t[k]) + r_1\sin(-\omega_1*t[k] + \phi_p); \\
\text{let } \{k \in 0..M-1\} y[0,k] & := r_0\sin(\omega_0*t[k]) + r_1\cos(-\omega_1*t[k] + \phi_p); \\
\text{let } \{k \in 0..M-1\} x[1,k] & := r_0\cos(\omega_0*t[k]) - r_1\sin(-\omega_1*t[k] + \phi_p); \\
\text{let } \{k \in 0..M-1\} y[1,k] & := r_0\sin(\omega_0*t[k]) - r_1\cos(-\omega_1*t[k] + \phi_p); \\
\text{let } \{k \in 0..M-1\} x[2,k] & := -r_0\cos(\omega_0*t[k]) + r_1\cos(-\omega_1*t[k] + \phi_m); \\
\text{let } \{k \in 0..M-1\} y[2,k] & := -r_0\sin(\omega_0*t[k]) + r_1\sin(-\omega_1*t[k] + \phi_m); \\
\text{let } \{k \in 0..M-1\} x[3,k] & := -r_0\cos(\omega_0*t[k]) - r_1\cos(-\omega_1*t[k] + \phi_m); \\
\text{let } \{k \in 0..M-1\} y[3,k] & := -r_0\sin(\omega_0*t[k]) - r_1\sin(-\omega_1*t[k] + \phi_m); \\
\text{solve;}
\end{align*}
\]
Limitations of the Model

• The integral gets discretized to a finite sum.

• Masses must be positive.

• Solutions can occur at local maxima and at saddle points. Looking only for local minima, we miss these.
Alternate Approach: Solve Equations of Motion

Instead of minimizing the action functional, which is an unconstrained optimization problem...

\[
\text{minimize } \int_0^{2\pi} \left( \sum_j \frac{1}{2} m_j \|\dot{z}_j\|^2 + \sum_{j,k:j<k} \frac{G m_j m_k}{\|z_j - z_k\|} \right) dt,
\]
subject to no constraints,

how about simply looking for trajectories that satisfy Newton’s laws:

\[
\text{minimize } 0,
\]
subject to \( m_j \ddot{z}_j^\alpha = - \sum_{k:k\neq j} G m_j m_k \frac{z_j^\alpha - z_k^\alpha}{\|z_j - z_k\|^3}, \quad j = 1, 2, \ldots, n, \quad \alpha = 1, 2. \)
AMPL Model for the Equations of Motion

var x {i in 0..N-1, j in 0..M-1} >= -5, <= 5;
var y {i in 0..N-1, j in 0..M-1} >= -5, <= 5;

var xdot {i in 0..N-1, j in 0..M-1} = (x[i,(j+1) mod M]-x[i,j])/dt;
var ydot {i in 0..N-1, j in 0..M-1} = (y[i,(j+1) mod M]-y[i,j])/dt;

var xdot2 {i in 0..N-1, j in 0..M-1} = (xdot[i,j]-xdot[i,(M+j-1) mod M])/dt;
var ydot2 {i in 0..N-1, j in 0..M-1} = (ydot[i,j]-ydot[i,(M+j-1) mod M])/dt;

minimize zero: 0;

subject to Fequalsma_x {i in 0..N-1, k in 0..M-1}:
    xdot2[i,k]
    = sum {j in 0..N-1: j != i} (x[j,k]-x[i,k]) / ((x[j,k]-x[i,k])^2+(y[j,k]-y[i,k])^2)^(3/2);

subject to Fequalsma_y {i in 0..N-1, k in 0..M-1}:
    ydot2[i,k]
    = sum {j in 0..N-1: j != i} (y[j,k]-y[i,k]) / ((x[j,k]-x[i,k])^2+(y[j,k]-y[i,k])^2)^(3/2);
NOTE: Action minimization found an orbit. But, it is immediately unstable as the middle figure shows.
Sensitivity Analysis

Let
\[ \xi^*(t) = \begin{bmatrix} x^*(t) \\ v^*(t) \end{bmatrix} \]
be a solution to
\[ \dot{\xi} = A(\xi) \]
where
\[ A \left( \begin{bmatrix} x(t) \\ v(t) \end{bmatrix} \right) = \begin{bmatrix} v(t) \\ a(x(t)) \end{bmatrix} \]
and
\[ a(x) = \begin{bmatrix} a_1(x) \\ \vdots \\ a_n(x) \end{bmatrix} \]
and
\[ a_j(x) = -\sum_{k:k\neq j} \frac{x_j - x_k}{\|x_j - x_k\|^2}, \quad j = 1, 2, \ldots, n. \]

Note: we’ve chosen units so that \( G = 1 \) and all masses are unit masses.
Consider a nearby solution $\xi(t)$:

$$
\dot{\xi}(t) = A(\xi(t)) \\
\approx A(\xi^*(t)) + A'(\xi^*(t))(\xi(t) - \xi^*(t)) \\
= \dot{\xi}^*(t) + A'(\xi^*(t))(\xi(t) - \xi^*(t)).
$$

Put $\Delta \xi = \xi - \xi^*$. Then $\dot{\Delta \xi} = A'(\xi^*(t))\Delta \xi$. A finite difference approximation yields

$$
\Delta \xi(t + h) = \Delta \xi(t) + \Delta t A'(\xi^*(t))\Delta \xi(t) \\
= (I + \Delta t A'(\xi^*(t))) \Delta \xi(t).
$$

Iterating around one period, we get:

$$
\Delta \xi(T) = \left( \prod_{i=0}^{n-1} (I + \Delta t A'(\xi^*(t_i))) \right) \Delta \xi(0),
$$

where $\Delta t = T/n$ and $t_i = i\Delta t$. 

Stability: all eigenvalues of

\[ \Lambda = \prod_{i=0}^{n-1} (I + \Delta t A'(\xi^*(t_i))) \]

must be at most one in magnitude.
<table>
<thead>
<tr>
<th>Name</th>
<th>$\max(\lambda_i(\Lambda))$</th>
</tr>
</thead>
<tbody>
<tr>
<td>Lagrange2</td>
<td>1.0000</td>
</tr>
<tr>
<td>Double Ellipse</td>
<td>1.0001</td>
</tr>
<tr>
<td>Triple Ellipse</td>
<td>1.0000</td>
</tr>
<tr>
<td>Quad Ellipse</td>
<td>5.7921</td>
</tr>
<tr>
<td>Ducati3</td>
<td>1.0000</td>
</tr>
<tr>
<td>Broucke-Henon</td>
<td>1.0000</td>
</tr>
<tr>
<td>Hill 2 months/yr</td>
<td>1.2430</td>
</tr>
<tr>
<td>Hill 3 months/yr</td>
<td>1.4187</td>
</tr>
<tr>
<td>Star of David</td>
<td>7.1140</td>
</tr>
<tr>
<td>DoubleDouble5</td>
<td>18.8167</td>
</tr>
<tr>
<td>DoubleDouble10</td>
<td>8.8401</td>
</tr>
<tr>
<td>DoubleDouble20</td>
<td>1.0000</td>
</tr>
<tr>
<td>FigureEight3</td>
<td>1.0000</td>
</tr>
<tr>
<td>Five Point Star</td>
<td>1.0000</td>
</tr>
<tr>
<td>Name</td>
<td>$\text{max}(\lambda_i(\Lambda))$</td>
</tr>
<tr>
<td>--------------------</td>
<td>-----------------------------------</td>
</tr>
<tr>
<td>Lagrange3</td>
<td>85.0138</td>
</tr>
<tr>
<td>Lagrange20</td>
<td>3.3644e+13</td>
</tr>
<tr>
<td>FigureEight4</td>
<td>168.38</td>
</tr>
<tr>
<td>FigureEight5</td>
<td>2126.0</td>
</tr>
<tr>
<td>PlateSaucer4</td>
<td>3658</td>
</tr>
<tr>
<td>1Month1Year</td>
<td>17.5830</td>
</tr>
<tr>
<td>1Month1YearAgain</td>
<td>1.7697</td>
</tr>
<tr>
<td>FoldedTriLoop4</td>
<td>7.4657e+04</td>
</tr>
<tr>
<td>Trefoil4</td>
<td>4.1726e+04</td>
</tr>
<tr>
<td>Braid4</td>
<td>7.7932e+03</td>
</tr>
<tr>
<td>Triangle</td>
<td>25.6724</td>
</tr>
<tr>
<td>Four Corners</td>
<td>269.0443</td>
</tr>
<tr>
<td>Binary Ellipse</td>
<td>1.4053</td>
</tr>
<tr>
<td>Hexagon</td>
<td>9.9829e+04</td>
</tr>
</tbody>
</table>
Making Stability an Objective

To the *Equations of Motion* model, add these perturbation parameters:

```plaintext
param dx := 1e-3;
param dy := 1e-3;
param dvx := 1e-3;
param dvy := 1e-3;
```

These new variables and objective function:

```plaintext
# perturbed trajectories
var xp {l in 0..4*N-1, i in 0..N-1, j in -1..M};
var yp {l in 0..4*N-1, i in 0..N-1, j in -1..M};
var xp_dot {l in 0..4*N-1, i in 0..N-1, j in -1..M-1} = (xp[l,i,j+1]-xp[l,i,j])/dt;
var yp_dot {l in 0..4*N-1, i in 0..N-1, j in -1..M-1} = (yp[l,i,j+1]-yp[l,i,j])/dt;
var xp_dot2 {l in 0..4*N-1, i in 0..N-1, j in 0..M-1} = (xp_dot[l,i,j]-xp_dot[l,i,j-1])/dt;
var yp_dot2 {l in 0..4*N-1, i in 0..N-1, j in 0..M-1} = (yp_dot[l,i,j]-yp_dot[l,i,j-1])/dt;

minimize instability:

```plaintext
sum {l in 0..N-1, j in 0..N-1: l != j} (xp[l,j,M-1] - x[j,M-1])^2
+ sum {j in 0..N-1} (xp[j,j,M-1] - dx-x[j,M-1])^2
+ sum {l in 0..N-1, j in 0..N-1: l != j} (yp[l,j,M-1] - y[j,M-1])^2
+ sum {j in 0..N-1} (yp[j,j,M-1] - dy-y[j,M-1])^2
+ sum {l in 0..N-1, j in 0..N-1: l != j} (xp_dot[l,j,M-1] - xdot[j,M-1])^2
+ sum {j in 0..N-1} (xp_dot[j,j,M-1] - dvx-xdot[j,M-1])^2
+ sum {l in 0..N-1, j in 0..N-1: l != j} (yp_dot[l,j,M-1] - ydot[j,M-1])^2
+ sum {j in 0..N-1} (yp_dot[j,j,M-1] - dvy-ydot[j,M-1])^2
```

And...
And these constraints defining the perturbed trajectories:

subject to F_eq_ma_x_pert {l in 0..4*N-1, i in 0..N-1, k in 0..M-1}:
    xp_dot2[l,i,k]
    = sum {j in 0..N-1: j != i}
    (xp[l,j,k]-xp[l,i,k]) / ((xp[l,j,k]-xp[l,i,k])^2+(yp[l,j,k]-yp[l,i,k])^2)^(3/2);

subject to F_eq_ma_y_pert {l in 0..4*N-1, i in 0..N-1, k in 0..M-1}:
    yp_dot2[l,i,k]
    = sum {j in 0..N-1: j != i}
    (yp[l,j,k]-yp[l,i,k]) / ((xp[l,j,k]-xp[l,i,k])^2+(yp[l,j,k]-yp[l,i,k])^2)^(3/2);

subject to x_pert_init {i in 0..N-1}: xp[i,i,0] = x[i,0] + dx;
subject to y_pert_init {i in 0..N-1}: yp[i,i,0] = y[i,0] + dy;

subject to xdot_pert_init {i in 0..N-1}: xp_dot[i,i,0] = xdot[i,0] + dvx;
subject to ydot_pert_init {i in 0..N-1}: yp_dot[i,i,0] = ydot[i,0] + dvy;
## Numerical Results

<table>
<thead>
<tr>
<th>Name</th>
<th>Obj. Func. Value</th>
</tr>
</thead>
<tbody>
<tr>
<td>Ducati3</td>
<td>1.4e-5</td>
</tr>
<tr>
<td>Star of David</td>
<td>1.6e-5</td>
</tr>
<tr>
<td>1Month1Year</td>
<td>1.4e-5</td>
</tr>
<tr>
<td>1Mini1Month1Year</td>
<td>5.5e-5</td>
</tr>
<tr>
<td>1Mini1Month1Year2</td>
<td>2.5e-1</td>
</tr>
<tr>
<td>1Mini1Month1Year(c)</td>
<td>4.2e-2</td>
</tr>
<tr>
<td>Five Point Star</td>
<td>8.9e-6</td>
</tr>
</tbody>
</table>
Leap-Frog Midpoint Integrator (using a Spring)

Differential equation:
\[ \ddot{x} = -x \]

Given: \( x(0), v(0) \)

Compute:
\[
\begin{align*}
    a(0) &= -x(0) \\
    v(h/2) &= v(0) + (h/2)a(0) 
\end{align*}
\]

For \( t = h, 2h, \ldots \)
\[
\begin{align*}
    a(t) &= -x(t) \\
    v(t + h/2) &= v(t - h/2) + ha(t) \\
    x(t + h) &= x(t) + hv(t + h/2) 
\end{align*}
\]

<table>
<thead>
<tr>
<th></th>
<th>( t )</th>
<th>( x )</th>
<th>( v )</th>
<th>( a )</th>
</tr>
</thead>
<tbody>
<tr>
<td>0.0</td>
<td>1.000</td>
<td>0.000</td>
<td>-1.000</td>
<td></td>
</tr>
<tr>
<td>0.1</td>
<td>0.995</td>
<td>-0.050</td>
<td>-0.995</td>
<td></td>
</tr>
<tr>
<td>0.2</td>
<td>0.980</td>
<td>-0.150</td>
<td>-0.980</td>
<td></td>
</tr>
<tr>
<td>0.3</td>
<td>0.955</td>
<td>-0.248</td>
<td>-0.955</td>
<td></td>
</tr>
<tr>
<td>0.4</td>
<td>0.921</td>
<td>-0.343</td>
<td>-0.921</td>
<td></td>
</tr>
<tr>
<td>0.5</td>
<td>0.877</td>
<td>-0.435</td>
<td>-0.877</td>
<td></td>
</tr>
</tbody>
</table>
The Midpoint Integrator

if (integrator == MIDPOINT) {
    for (j=0; j<n; j++) {
        p[j].x += p[j].vx * dt;
        p[j].y += p[j].vy * dt;
    }
    for (j=0; j<n; j++) {
        p[j].ax = 0; p[j].ay = 0;
        for (i=0; i<n; i++) {
            if (i != j) {
                double r3 = dist3(p[i], p[j]);
                if (r3<r03) r3=r03;
                p[j].ax -= G * p[i].m * (p[j].x - p[i].x)/r3;
                p[j].ay -= G * p[i].m * (p[j].y - p[i].y)/r3;
            }
        }
    }
    for (j=0; j<n; j++) {
        p[j].vx += p[j].ax * dt;
        p[j].vy += p[j].ay * dt;
    }
}
Backup Slides
\[ \delta^2 A = \int_0^{2\pi} \sum_j \sum_{\alpha} \left( \dot{\delta z}_j^\alpha \right)^2 \, dt \]

\[ + 3 \int_0^{2\pi} \sum_{j,k: j \neq k} \sum_{\alpha, \beta} \frac{(z_j^\alpha - z_k^\alpha)(\dot{z}_j^\beta - \dot{z}_k^\beta)(\delta z_j^\beta - \delta z_k^\beta)}{\| z_j - z_k \|^5} \delta z_j^\alpha \, dt \]

\[ - \int_0^{2\pi} \sum_{j,k: j \neq k} \sum_{\alpha} \frac{\delta z_j^\alpha - \delta z_k^\alpha}{\| z_j - z_k \|^3} \delta z_j^\alpha \, dt \]
Lagrange 20 Bodies
Double Ellipse
Triple Ellipse
Hill – 2 Months/Year
Hill – 3 Months/Year
Figure Eight 4 Bodies
Figure Eight 5 Bodies
Double/Double – 5 Months/Year
Double/Double – 10 Months/Year
Double/Double – 20 Months/Year
Plate and Saucer
Ortho Quasi Ellipse
Folded TriLoop

4 Bodies
Star of David 4 Bodies
Triangle 4 Bodies
Broucke-Henon (aka Ducati) 3 Bodies
Hill 2 Months/Year
Five Point Star

4 Bodies
Hexagon